
Abstract In this paper we present a language for Abstract State Machine speci-
fications. The ASM metamodel (AsmM), introduced in [18], is aMOF-compliant
metamodel representing in an abstract and visual way the concepts and constructs
of the ASMs formalism as described in [3]. Here we present aconcrete syntax
(AsmM-CS), an EBNF (extended Backus-Naur Form) grammar derived from the
AsmM as a textual notation to be used by modelers to effectively write ASM
models complaint with AsmM. We also give an overview of the technique ap-
plied to derive AsmM-CS from AsmM, showing how the OMG metamodel-based
approach can be exploited to derive languages from metamodels.

1 Introduction

Metamodelling is nowadays supported by the Model-Driven Architecture (MDA) [11]
approach as a modular and layered way to endow a language withanabstract notation,
so separating the abstract syntax and semantics of the language constructs from their
differentconcrete notations.In the MDA vision, a language has to be equipped by at
least a proper metamodel-based definition of the abstract syntax of the language, an easy
to learn concrete syntax, possibly graphic, a well-foundedsemantics, and a uniform
style (through, e.g., the XML base format [25]) of representing language constructs for
interchanging purposes.

As already discussed in [18], a standard interchange formatis of particular interest
for the ASM (Abstract State Machines) community, since ASM tools have been usually
developed by individual research groups, are loosely coupled and have syntaxes strictly
depending on the target environment (compare, for example,AsmGofer [19], ASM-SL
[6], XASM [1], ASML [14]).

Taking advantage of the metamodel-based approach, we defined in [18] a new
metamodel, calledAbstract State Machine Metamodel(AsmM, in brief), for the ASM
method. The AsmM framework introduces anabstract syntax– a MOF-compliant meta-
model [12] representing in an abstract and visual way the concepts and constructs of
the ASM formalism as described in [3] –, and aninterchange syntax– a standard XMI-
based format [24] automatically derived from the AsmM – for the interchange of ASM
models. Here, we present aconcrete syntax(AsmM-CS), namely an EBNF (extended
Backus-Naur Form) grammar derived from the AsmM as a textualnotation to write

ASM models in a textual form. For theAsmM semantics, we assume the ASM seman-
tics given in [3]. A complete description of the AsmM 1.0 specification is given in [2].

The AsmM proposal can be viewed as the first step towards a definition of a stan-
dardfamily of languagesfor the ASM formal method and a systematic integration of a
number of loosely-coupled ASM tools based upon metamodelling techniques.

We like to remark that the effort of developing a new BNF grammar for an ASM lan-
guage from scratch would not be less than the effort of developing a MOF-metamodel
for ASMs and deriving a BNF grammar from it. Moreover, the metamodel approach has
the advantage of being suitable to derive different BNF-like grammars from the same
MOF-metamodel, and benefits of some other features as tool support, abstract graphical
view, interchange format, standard encoding in programming languages, etc.

The paper is organized as follows. In Section 2, the AsmM concrete syntax is pre-
sented. Section 3 shows an example of an ASM specification written using our concrete
syntax. The process of deriving the AsmM-CS from the AsmM is described in Section
4. Related and future work are given in Sections 5 and 6.

2 The AsmM Concrete Syntax (AsmM-CS)

To understand the content of the following sections, we assume the reader to be familiar
with the AsmM. The AsmM concrete syntax can be divided roughly into four parts,
namely thestructural language, thedefinitional language, the language of terms, and
the behavioural language (orlanguage of rules). The structural language corresponds to
theASMStructure package of the AsmM abstract syntax and provides the constructs
to describe the structure of an ASM. The definitional language corresponds to theASM-
Definitions package of the AsmM abstract syntax and provides the notation to define
the basic ASM elements such as functions, domains, rules, and axioms. The language of
terms corresponds to theASMTerms package of the AsmM abstract syntax and provides
all kinds of syntactic expressions which can be evaluated inan ASM state. Finally, the
behavioural language corresponds to theASMTransitionRules package of the AsmM
abstract syntax and provides a notation to specify the ASM transition rule schemes.

The following sections present the notation of each syntactic category in a tabular
way for better readability. The complete EBNF grammar can befound in [2].

We adopt the following conventions: keywords appear inbold face; a pair of square
braces [] (not in bold face) indicates that the enclosed expression is optional and must
not be considered part of the concrete notation; the notation t1, ...,tn indicates one or
more elements.

2.1 The Structural Language

An ASM model is structured into four sections: aheader, aninitialization, abodyand a
main rule. Figure 1 shows the concrete syntax for each section of an ASMmodel. The
name of the ASM is specified before the header section together with the optional key-
wordisAsyncr to specify if the ASM is an asynchronous multi-agent or not;isAsyncr
has no meaning for single-agent ASMs. A lightweight notion of module is also sup-
ported; an ASMmoduleis an ASM without a main rule and without a characterization



of the set of initial states. A module has the same syntax of anASM with the keyword
asm replaced by the keywordmodule.

The header section consists of someimport clausesand oneexport clausewhich
describe the ASM interface for the communication with otherASMs or ASM mod-
ules. Thesignaturecontains thedeclarationsof domains and functions. Every ASM is
allowed to use only identifiers (for domains, functions and rules) which are declared
within its header’s signature or imported from other modules. The imported domains
and functions will be statically added in the signature of the machine as new functions
and the imported rules will enrich the module interface of the machine.

The initialization section consists of a set ofinitial states, one of which is elected as
default. Figure 1 shows the schema of an initial state. An initial state defines an initial
value for each dynamic function and eachconcrete-domain(see sect. 2.2 below) already
declared in the signature of the ASM1. In addition, the initial state associates eachagent
of the machine with the agentid name andprogram(a named transition rule).

The body section consists ofdefinitionsof static concrete-domains and static/derived
functions already declared in the signature2, declarations(definitions) of transition
rules, anddeclaration(definitions) of axioms stating assumptions and constraints on
functions, domains, and transition rules of the ASM.

The main rule is a named transition rule denoted by the keyword main. It is closed
(namely it does not contain free variables) so that its semantics depends only on the
state of the machine. Executing an ASM means executing its main rule starting from a
specified initial state. If the ASM has no main rule, by default, the ASM starts executing
in parallel the agent’s programs given by the agent initialization clauses in a specified
initial state.

2.2 The Definitional Language

To declare an ASM function it is necessary to specify its name, domain, and codomain.
The function name must be preceded by one of the keywordsstatic, dynamicor derived,
depending on the function type. Dynamic functions are further classified inmonitored,
controlled, shared, out, andlocal. Dynamic functions are allowed to be declared aslocal
only in the scope of a turbo rule with local state (see sect. 2.4). They are not considered
to be part of the signature of the ASM. Figure 2 shows the concrete syntax for a function
declaration.

Our language admits the following domain (or universe) classification:type-domains
and concrete-domains. The type-domains represent all possiblesuper domains3 and
are further classified in:basic type-domains, domains for primitive data values like
booleans, reals, integers, naturals, strings, etc.;structured type-domains, domains for
building data structures (like sets, sequences, bags, maps, tuples, etc.) over other type-
domains;abstract type-domain, dynamic user-named domains whose elements have no
precise structure and are imported as fresh elements from a possibly infinite reserve

1 Only dynamic functions and dynamic concrete-domains need to be initialized.
2 Only static/derived functions and static concrete-domains need to be defined.
3 In practical applications, the superuniverse|S| of an ASM stateSis usually divided into smaller

universes. In the AsmM, these smallersuper domainsare calledtype-domains.



Figure 1. Structure of an ASM model



Figure 2. Notation for declaring a functionf from D toC

by means ofextend rules(see sect. 2.4 below); andenum domains, finite user-named
enumerations to introduce new concepts of type (e.g. one maydefine the enumeration
Color = {RED,GREEN,BLUE}to introduce the new concept of “color”).

Concrete domains are, instead, user-named sub-domains of type-domains. As for
functions, a concrete domain can be static or dynamic.

Figure 3. Notation for declaring a domain D

Figure 3 shows the notation for declaring a domain. As basic type-domains (defined
in the standard library) we admit only:Complex, Real, Integer, Natural, String,
Char, Boolean, Rule, and the singletonUndef. Moreover, two other special abstract
type-domains are considered predefined: theAgent domain for agents, and theReserve
domain for representing a possibly infinite reserve which provides fresh elements to
increase the working space of an ASM. TheReserve domain is considered abstract,



and thereforedynamic, since it is updated automatically upon execution of anextend
rule (see sect. 2.4 below) – it can not be updated directly by othertransition rules –.

Finally, Figure 4 shows named transition rules and axioms declaration.

Figure 4. Notation for declaring rules and axioms

2.3 The Language of Terms

Terms are syntactic objects denoting elements of ASM states. Terms are interpreted in
an ASM stateS if the elements of the super-universe ofS– the set of interpretation of
function names of the ASM signature – are assigned to the variables of the terms.

As in first-order logic, we admit basic terms (variables, constants, and function ap-
plications) and in addition we introduce special terms liketuples, collection terms (sets,
maps, sequences, and bags), variable-binding terms (let-terms, comprehension terms,
finite quantification terms, etc.), etc. These last special terms have been added by bor-
rowing concepts from the ASM-SL language [6].

The termRuleAsTermis a special term, arule term, used to represent a transition
rule where a term is expected (e.g as actual parameter in a rule application to represent
a transition rule). Its interpretation results, therefore, in a transition rule.

As in [3], we classify variables (not to be confused with 0-ary functions fixed by the
ASM signature) inlogical variables, location variablesandrule variables. A location
variable, appearing as formal parameter in a rule declaration, can be used in the rule
body on the left-hand side of an update rule, while a rule variable can be used at places
where a transition rule is expected. All other variables areconsideredlogical. In a rule
application, logical variables have to be replaced by generic terms; location variables
have to be replaced by location terms – namely, function terms which start with a dy-
namic function fixed by the ASM signature –; and rule variables have to be replaced by
rule terms.

Within the transition rules, each agent can identify itselfby means of a special
reserved 0-ary functionself : Agent, which is interpreted by each agentaasa. Moreover,



Figure 5. Basic Terms Notation

for a functionf defined fromX toY, both the expressionsf (self,x) andself. f (x) denote
the private version off (x) of the agentself.

2.4 The Behavioral Language (or the Language of Rules)

We classify transition rules in two groups:basic rules andturbo rules. According to
[3], the former are simply rules, like theskip ruleand theupdate rule, while the latter
are rules, like thesequence ruleand theiterate rule, introduced to support practical
composition and structuring principles of the ASMs. Other rule schemes are derived
from the basic and the turbo rules.

Figures 7, 8, and 9 show the concrete notation of all kinds of rules presented in [3].

Variables appearing in rules such aslet, forall andchooseare not free variable oc-
currences, but they are bound to thescopedetermined by the rule portion in which they
are used. These variables are not stored in the state of the ASM but in a local envi-
ronment. The formal parameters specified in a rule declaration are, therefore, the only
freely occurring variables in the rule body.

We consider anamed rule applicationa form of rule. We distinguish, in particular, a
macro rule application r[t1, ..,tn] (a macro-call rule) – resulting in the execution of the
expansion ofr obtained replacing in thebodypart of r the formal parametersx1, ..,xn

(of the declaration ofr) with the corresponding values of the actual parameterst1, ..,tn
– from aturbo rule application r(t1, ..,tn) (theturbo-call rule) with a turbo submachine
semantics (see sect. 4.1.2 of [3]) to express in abstract form the usual imperative calling



Figure 6. Extended Terms Notation



Figure 7. Basic Rules Notation



Figure 8. Turbo Rules Notation

Figure 9. Derived Rules Notation



mechanism. Moreover, since it is assumed that turbo rule applications have acall by-
namesemantics, i.e. the formal parameters are substituted in the rulebodyby the actual
parameters so that these are evaluated only later when they are used, acall-by-value
evaluation of a rule application can be achieved (as suggested in sect. 4.1.2 of [3])
combining the ruler with a let ruleas follows:

r(y1, . . . ,yn) = let (x1 = y1, . . . ,xn = yn) in body endlet

A turbo-return ruleis a mechanism which allows one to retrieve the intended return
value of a named turbo application ruler from a locationl determined by the rule
caller. Semantically speaking, a turbo-return rule denotes a rule with the overall effect
of executing the body ofr, where the 0-ary dynamic function denoted byl has been
substituted for a reserved 0-ary functionresultwhich acts as placeholder where to store
the intended return value. A good encapsulation disciplinewill take care, therefore, that
r (i) contains an update rule of the formresult:= t, and it (ii) does not modify the values
of terms appearing inl , since they contribute to determine the location where the caller
expects to find the return value.

3 An Example of an ASM Specification

In this section, we show the ASM specification of a Flip-Flop.The original model was
given at page 47 of [3] by means of the following two rules: thefirst one (FSM) which
models a generic finite state machine and the second one whichinstantiates the FSM for
a Flip-Flop.

FSM(i,cond,rule,j)= if ctl_state= i and condthen { rule, ctl_st:=j}
FLIPFLOP = {FSM(0,high,skip,1),FSM(1,low,skip,0)}

The AsmM specification of the Flip-Flop follows4:

asm flip_flop
signature:
domain State subsetof Integer
dynamic controlled ctl_state : State
dynamic monitored high : Boolean
dynamic monitored low : Boolean

default init initial_state:
function ctl_state = 0
function high = false
function low = false

4 We assume the following rules to distinguish among names of variables, enumeration ele-
ments, domains, rules, and functions: a variable identifierstarts always with an initial “$”; an
enum literal is a string of length greater than or equal to twoand consisting of upper-case let-
ters only; a domain identifier begins always with an upper-case letter; a rule identifier always
begins with the lower-case letter “r” followed by “_”; a function identifier always begins with
a lower-case letter, but can not start with “r_”.



definitions:
domain State = {0,1}
rule r_Fsm($i in State,$cond in Boolean, $rule in Rule,$j in State)=

if ctl_st = $i and $cond
then par

$rule
ctl_st := $j

endpar
endif

axiom over high(Boolean),low(Boolean): not (high and low)

main rule r_flip_flop =
par

r_Fsm(0,high,< <skip> >,1)
r_Fsm(1,low,< <skip> >,0)

endpar

4 How to Derive an AsmM-CS from the AsmM: Mapping MOF to
EBNF

MOF is a large OO modelling language with rich concepts to express information mod-
els. The main MOF modelling constructs are:Package, for containment of classes and
associations;Class, which contains attributes and participates in associations;Associ-
ation, which represents a set of links between instances of two specified classes and
which can have aggregation and composition properties;Attribute, which can be either
in the form of one of a range of data types or an instance of a class; andReference,
which is a class’s view on an association in which it participates. For more details on
these and other MOF modelling concepts, see the specification [12].

A mapping from MOF-based metamodels to EBNF grammars (forward engineer-
ing) is more demanding than the opposite (reverse engineering). The reason is that
MOF-based metamodels inherently contain more informationthan EBNF grammars.
An EBNF grammar can be presented as a tree of nodes and directed edges, but the
edges themselves do not contain as much information as properties in a metamodel.
Metamodels instead are graphs with special edges that can beinterpreted in many ways
(generalization semantics, aggregation semantics, composition semantics, etc.). A map-
ping from EBNF grammars to metamodels uses only a subset of the capabilities of
metamodels, and the generated metamodel may need to be further enriched in order to
make it more abstract.

The AsmM-CS has been derived from the semantic interpretation of the AsmM in
order to simplify the mapping between the concrete syntax and the abstract one5. In
our AsmM we used a subset of the MOF 1.4 constructs which is small enough to be

5 This approach was inspired by the recent OCL 2.0 specification [13] (which is also based on a
MOF-compliant metamodel) made to align the OCL language with respect to UML/MOF 2.0
[23].



easily translated in EBNF. Table 1 describes the subset of the MOF constructs we used
together with the mapping rules from MOF to EBNF we adopted toobtain the AsmM-
CS. Figure 10 shows the application of the mapping rules to a fragment of AsmM.

MOF Concepts EBNF Concepts

Package –
Class A non terminal C. The production rule for C is determined by the

attributes and the relations with other classes.
Attribute (instance level, single primitive value)

of Boolean type A special keyword reflecting the name of the attribute followed by ?
of Enum type A choice group of keywords reflecting the names of the enum literals
of String type Ientifiers

Association see Reference
Generalization

from a concrete
super-class

The production rules for the non terminals of the super-class and of the
sub-classes are determined as usual. The properties inherited have
the same representation in both the production for the non terminal of
the super-class and the productions for the non terminals of the
sub-classes.

from an abstract
super-class A to
sub-classes A1, . . . ,An

A choice group A ::= A1| . . . |An

The properties inherited have the same representation in all
productions of the non terminals.

Aggregation see Reference
Composition see Reference
Reference

in a composition or
aggregation

A full representation, i.e. an occurrence of the non terminal of the class
of the contained instance.

in a simple association A keyword combined with either a full representation of the instance
(an occurrence of the non terminal), or a representation by name, i.e.
an occurrence of the identifier of the instance.

Multiplicity Repetition ranges
0..* *
1..* +
0..1 ? or []

Table 1.Mapping from MOF to EBNF

MOF Referencesare a means for classes to be aware of class instances that play a
part in an association (either simple or composite), by providing a view into the associ-
ation as it pertains to the observing instance. For this reason, the representation within
the production rule of a non terminal corresponding to a class instance of a reference
depends in part on the nature of the association to which it refers.

Since in a composition the contained instance does not existoutside the scope of
the whole instance, the reference to the contained instanceis represented in the produc-
tion rule of the non terminal corresponding to the whole class by afull representation,



Figure 10.From MOF to EBNF: an example

i.e. as a non terminal (corresponding to the class of the contained instance) combined
with other parts of the production taking into account the multiplicities. An eventual
reference to the whole instance is not represented.

In a simple association (that is, the associated instance can exist outside the scope
of the other instance), instead, the representation of the reference consists of a keyword
reflecting the name of the reference (or the name of the association end) combined with
either a full representation of the instance (an occurrenceof the non terminal of the
associated class), orby name, i.e. an occurrence of the identifier (the name attribute
value of the instance) if any, taking into account the multiplicities.

The grammar obtained, of course, may need to be further optimized and concretized
for the purpose of construction of a parser/compiler. Respect to the derivation process
presented here, for example, the AsmM-CS has been already customized a bit, in or-
der to allow alternative representations of the same concepts (i.e. a class instance in
the metamodel can admit many equivalent notations) such as the interval notation for
sets/sequences/bags of reals, special expressions to support the infix notation for some
well-known functions on basic domains (like plus, minus, mult, etc.), the use of the two
keywordsasm andmodule to better distinguish an ASM from an ASM module, etc..

5 Related Work

Concerning the definition of a concrete language for ASMs, other previous proposals
exist. The Abstract State Machine Language (AsmL) [14] developed by the Foundation
Software Engineering group at Microsoft is the greatest effort in this respect. AsmL is
a rich executable specification language, expression and object oriented, based on the



theory of Abstract State Machines and fully integrated intothe .NET framework and
Microsoft development tools; AsmL does not provide a semantic structure targeted for
the ASM method. “One can see it as a fusion of the Abstract State Machine paradigm
and the .NET type system, influenced to an extent by other specification languages like
VDM or Z” [26]. Adopting a terminology currently used in the MDA [11] approach,
AsmL is a powerful platform-specific modelling language (PSM) for the .NET type
system. A similar consideration can be made also for the AsmGofer language [19]. An
AsmGofer specification can be seen, in fact, as a specific PSM for the Gofer environ-
ment. Other specific languages for the ASMs, no longer maintained, are ASM-SL [6],
which adopts a functional style being developed in ML and which has inspired us in
the language of terms, and XASM [1] which is integrated in Montages, an environment
generally used for defining programming language semanticsand grammar.

A platform-independent modelling (PIM) language for ASMs could allow to de-
fine precisetransformation bridgesin order to automatically map an ASM PIM into an
AsmGofer-PSM, or into an AsmL-PSM, and so on. In the same manner, we may “com-
pile” ASMs models into programming languages such as C++, C#, Java and so on, to
provide efficient code generation capabilities andround trip engineeringfacilities as
well.

Concerning the metamodelling technique for the definition of languages, we can
mention the official metamodels supported by the OMG [15] forMOF itself [12], for
UML [22], for OCL [13], for CWM [4], etc. Academic communities like the Graph
Transformation community [21,16] and the Petri Net community [17,7], have also
started to settle their tools on general metamodels and XMI/XML-based formats.

Recently, a metamodel for the ITU language SDL-2000 [20] wasdeveloped [9,10].
The authors present a semi-automaticreverse engineeringmethodology that allows the
derivation of a metamodel from a formal syntax definition of an existing language.
The SDL metamodel has been derived from the SDL grammar usingthis methodology.
Their method is complementary with the derivation process presented in this paper (see
Sect. 4 above), since our approach has to be considered aforward engineeringprocess
consisting in deriving a concrete textual notation from an abstract metamodel.

Other more complex MOF-to-text tools, capable of generating text grammars from
specific MOF-based repositories, exist [8,5]. These tools render the content of a MOF-
based repository (known as a MOFlet) in textual form, conforming to some syntactic
rules (grammar). However, although automatic, since they are designed to work with
any MOF model and generate their target grammar based on predefined patterns, they
do not permit a detailed customization of the generated language.

6 Conclusions and Future Directions

Within the language engineering area, metamodels provide astandardized visual rep-
resentation easy to learn and supported by a number of tools to design, implement and
document languages.

We propose a metamodelling-based definition of a language, the AsmM, for the
ASM theory described in [3]. The AsmM has been defined using two metalanguages:



the MOF [12] and EBNF. The MOF is used to describe the abstractsyntax, while the
EBNF is used to describe the (textual) concrete syntax.

A prior version of the AsmM and also of an XML/XMI -based interchange syntax
was presented in [18]. In this paper, we present the AsmM concrete syntax and we
describe the MOF-to-EBNF mapping rules applied to derive the concrete syntax from
the abstract syntax.

The all AsmM is still evolving because, in order to make it a “standard”, one of our
main goals is to modify existing constructs and add (or even remove) concepts to meet
the needs of the ASM community. We also plan to develop a parser for the AsmM lan-
guage based on our syntax definition to link it to a proper modelling environment, and
to provide support for “transformations” (text-To-MOF, MOF-To-text, XMI-To-MOF,
MOF-To-XMI, MOF-To-ANY, etc.). In this way, a designer could write her/his ASM
specification in the textual form of AsmM-CS, transform it inthe XMI interchange
format easily readable and modifiable by tools.

Benefits provided by a standardized notation for the ASMs maycontribute to in-
crease the practical use of the ASMs method and provide an efficient interaction among
ASM tools for a higher quality design based on the ASM formalism.

Appendix A – the AsmM-CS

In the EBNF grammar reported below, nonterminals are plain and literal symbols are
enclosed in double quotes. In addition, words enclosed in angle brackets indicate a
placeholder for a literal value that must be substituted with an actual value ( e.g.,
<DIGIT> ::= [0" - "9"]").

The structural language

Asm ::= ( "asynchr" )? ( "asm" | "module"
) <ID> Header (( Initialization )* "default" Initialization (
Initialization )*)? Body ( "main" RuleDeclaration )?

Header ::= ( ImportClause )* ( ExportClause )? Signature

ImportClause ::= "import" <MOD_ID> ( "(" ( <ID_DOMAIN> |
<ID_FUNCTION> | <ID_RULE> ) ( "," ( <ID_DOMAIN> | <ID_FUNCTION> |
<ID_RULE> ) )* ")" )?

ExportClause ::= "export" ( ( <ID_DOMAIN> | <ID_FUNCTION> |
<ID_RULE> ) ( "," ( <ID_DOMAIN> | <ID_FUNCTION> | <ID_RULE> ) )*
) | "*"

Signature ::= "signature" ":" ( Domain )* ( Function )*

Initialization ::= "init" <ID> ":" ( DomainInitialization )*
( FunctionInitialization )*
( AgentInitialization )*

FunctionInitialization ::= "function" <ID_FUNCTION> ( "("
VariableTerm "in" ( <ID_DOMAIN> | StructuredTD ) ( ","
VariableTerm "in" ( <ID_DOMAIN> | StructuredTD ) )* ")" )? "="
Term

DomainInitialization ::= "domain" <ID_DOMAIN> "=" Term

AgentInitialization ::= "<" ID_AGENT ">" ":" "agent" "<" (
<ID_RULE> | Rule ) ">"

Body ::= "definitions" ":" ( DomainDefinition )*
( FunctionDefinition )*
( RuleDeclaration )*
( Axiom )*

FunctionDefinition ::= "function" <ID_FUNCTION> ( "(" VariableTerm
"in" ( <ID_DOMAIN> | StructuredTD ) ( "," VariableTerm "in" (
<ID_DOMAIN> | StructuredTD ) )* ")" )? "=" Term

DomainDefinition ::= "domain" <ID_DOMAIN> "=" Term



RuleDeclaration ::= "rule" ( "macro" )? <ID_RULE> ( "("
VariableTerm "in" ( <ID_DOMAIN> | StructuredTD ) ( ","
VariableTerm "in" ( <ID_DOMAIN> | StructuredTD ) )* ")" )? "="
Rule

Axiom ::= "axiom" "over" ( ID_DOMAIN | ( ID_FUNCTION "(" (
<ID_DOMAIN> | StructuredTD )? ")" ) | ID_RULE ) ( "," ( ID_DOMAIN
| ( ID_FUNCTION "(" ( <ID_DOMAIN> | StructuredTD )? ")" ) |
ID_RULE ) )* ":" Term

The definitional language

Domain ::= (ConcreteDomain | TypeDomain)

ConcreteDomain ::= ( "dynamic" )? "domain" <ID_DOMAIN> "subsetof"
( <ID_DOMAIN> | StructuredTD )

TypeDomain ::= ( StructuredTD | EnumTD | AbstractTD | BasicTD |
"anydomain" <ID_DOMAIN> )

BasicTD ::= "basic" "domain" <ID_DOMAIN>

AbstractTD ::= "abstract" "domain" <ID_DOMAIN>

EnumTD ::= "enum" "domain" <ID_DOMAIN> "=" "{" EnumElement ( "|"
EnumElement )* "}"

EnumElement ::= <ID_ENUM>

StructuredTD ::= ( ProductDomain | SequenceDomain | PowersetDomain
| BagDomain | MapDomain )

ProductDomain ::= "Prod" "(" ( <ID_DOMAIN> | StructuredTD ) ( ","
( <ID_DOMAIN> | StructuredTD ) )+ ")"

SequenceDomain ::= "Seq" "(" ( <ID_DOMAIN> | StructuredTD ) ")"

PowersetDomain ::= "Powerset" "(" ( <ID_DOMAIN> | StructuredTD )
")"

BagDomain ::= "Bag" "(" ( <ID_DOMAIN> | StructuredTD ) ")"

MapDomain ::= "Map" "(" ( <ID_DOMAIN> | StructuredTD ) "," (
<ID_DOMAIN> | StructuredTD ) ")"

Function ::= ( BasicFunction | DerivedFunction )

BasicFunction ::= ( StaticFunction | DynamicFunction )



DerivedFunction ::= "derived" <ID_FUNCTION> ":" ( ( <ID_DOMAIN> |
StructuredTD ) "->" )? ( <ID_DOMAIN> | StructuredTD )

StaticFunction ::= "static" <ID_FUNCTION> ":" ( ( <ID_DOMAIN> |
StructuredTD ) "->" )? ( <ID_DOMAIN> | StructuredTD )

DynamicFunction ::= ( "dynamic" )? ( "monitored" | "controlled" |
"shared" | "out" | "local" ) <ID_FUNCTION> ":" ( ( <ID_DOMAIN> |
StructuredTD ) "->" )? ( <ID_DOMAIN> | StructuredTD )

The language of terms

Term ::= ( Expression | ExtendedTerm )

Expression ::= or_xorLogicExpr ( "implies" or_xorLogicExpr | "iff"
or_xorLogicExpr )*

or_xorLogicExpr ::= andLogicExpr ( "or" andLogicExpr | "xor"
andLogicExpr )*

andLogicExpr ::= notLogicExpr ( "and" notLogicExpr )*

notLogicExpr ::= ( "not" includesExpr | includesExpr )

includesExpr ::= relationalExpr ( "in" relationalExpr | "notin"
relationalExpr )?

relationalExpr ::= additiveExpr ( ( "=" additiveExpr | "!="
additiveExpr | "<" additiveExpr | "<=" additiveExpr | ">"
additiveExpr | ">=" additiveExpr ) )*

additiveExpr ::= multiplicativeExpr ( "+" multiplicativeExpr | "-"
multiplicativeExpr )*

multiplicativeExpr ::= powerExpr ( "*" powerExpr | "/" powerExpr |
"mod" powerExpr )*

powerExpr ::= unaryExpr ( "^" unaryExpr )* unaryExpr ::= ( "-" "("
basicExpr ")" | basicExpr )

basicExpr ::= ( BasicTerm | DomainTerm | "(" Expression ")" )

BasicTerm ::= ( ConstantTerm | VariableTerm | FunctionTerm )

FunctionTerm ::= ( ID_AGENT "." )? <ID_FUNCTION> ( TupleTerm )?

LocationTerm ::= ( ID_AGENT "." )? <ID_FUNCTION> ( TupleTerm )?

VariableTerm ::= <ID_VARIABLE>



ConstantTerm ::= ( ComplexTerm | RealTerm | IntegerTerm |
NaturalTerm | CharTerm | StringTerm | BooleanTerm | UndefTerm |
EnumTerm )

ComplexTerm ::= <COMPLEX_NUMBER>

RealTerm ::= ( "+" | "-" )? <REAL_NUMBER>

IntegerTerm ::= ( "+" | "-" ) <NUMBER> NaturalTerm ::= <NUMBER>

CharTerm ::= <CHAR_LITERAL> StringTerm ::= <STRING_LITERAL>

BooleanTerm ::= ( "true" | "false" ) UndefTerm ::= "undef"

EnumTerm ::= <ID_ENUM>

ExtendedTerm ::= ( ConditionalTerm | CaseTerm | TupleTerm |
CollectionTerm | VariableBindingTerm | RuleAsTerm | DomainTerm )

ConditionalTerm ::= "if" Term "then" Term ( "else" Term )? "endif"

CaseTerm ::= "switch" Term ( "case" Term ":" Term )+ ( "otherwise"
Term )? "endswitch"

TupleTerm ::= "(" Term ( "," Term )* ")"

CollectionTerm ::= ( SequenceTerm | SetTerm | MapTerm | BagTerm )

SequenceTerm ::= "[" ( Term ( ( "," Term )+ | ( ".." Term ( ","
( <REAL_NUMBER> | <NUMBER> ) )? ) )? )? "]"

SetTerm ::= "{" ( Term ( ( "," Term )+ | ( ".." Term ( "," (
<REAL_NUMBER> | <NUMBER> ) )? ) )? )? "}"

MapTerm ::= "{" ( "->" | ( Term "->" Term ( "," Term "->" Term )*
) ) "}"

BagTerm ::= "<" ( Term ( ( "," Term )+ | ( ".." Term ( "," (
<REAL_NUMBER> | <NUMBER> ) )? ) )? )? ">"

VariableBindingTerm ::= ( LetTerm | FiniteQuantificationTerm |
ComprehensionTerm )

FiniteQuantificationTerm ::= ( ForallTerm | ExistUniqueTerm |
ExistTerm )

ExistTerm ::= "(" "exist" VariableTerm "in" Term ( ","
VariableTerm "in" Term )* ( "with" Term )? ")"



ExistUniqueTerm ::= "(" "exist" "unique" VariableTerm "in" Term (
"," VariableTerm "in" Term )* ( "with" Term )? ")"

ForallTerm ::= "(" "forall" VariableTerm "in" Term ( ","
VariableTerm "in" Term )* ( "with" Term )? ")"

LetTerm ::= "let" "(" VariableTerm "=" Term ( "," VariableTerm "="
Term )* ")" "in" Term "endlet"

ComprehensionTerm ::= ( SetCT | MapCT | SequenceCT | BagCT )

SetCT ::= "{" Term "|" VariableTerm "in" Term ( "," VariableTerm
"in" Term )* ( "with" Term )? "}"

MapCT ::= "{" Term "->" Term "|" VariableTerm "in" Term ( ","
VariableTerm "in" Term )* ( "with" Term )? "}"

SequenceCT ::= "[" Term "|" VariableTerm "in" Term ( ","
VariableTerm "in" Term )* ( "with" Term )? "]"

BagCT ::= "<" Term "|" VariableTerm "in" Term ( "," VariableTerm
"in" Term )* ( "with" Term )? ">"

DomainTerm ::= ( <ID_DOMAIN> | StructuredTD )

RuleAsTerm ::= "<<" Rule ">>"

The language of rules

Rule ::= ( BasicRule | TurboRule | TermAsRule | DerivedRule )

TermAsRule ::= ( FunctionTerm | VariableTerm )

BasicRule ::= ( SkipRule | UpdateRule | MacroCallRule | BlockRule
| ConditionalRule | ChooseRule | ForallRule | LetRule | ExtendRule
)

SkipRule ::= "skip"

UpdateRule ::= ( LocationTerm | VariableTerm ) ":=" Term

BlockRule ::= "par" Rule Rule ( Rule )* "endpar"

ConditionalRule ::= "if" Term "then" Rule ( "else" Rule )? "endif"

ChooseRule ::= "choose" VariableTerm "in" Term ( "," VariableTerm
"in" Term )* "with" Term "do" Rule ( "ifnone" Rule )?

ForallRule ::= "forall" VariableTerm "in" Term ( "," VariableTerm



"in" Term )* "with" Term "do" Rule

LetRule ::= "let" "(" VariableTerm "=" Term ( "," VariableTerm "="
Term )* ")" "in" Rule "endlet"

MacroCallRule ::= <ID_RULE> "[" ( Term ( "," Term )* )? "]"

ExtendRule ::= "extend" <ID_DOMAIN> "with" VariableTerm ( ","
VariableTerm )* "do" Rule

TurboRule ::= ( SeqRule | IterateRule | TurboReturnRule |
TurboCallRule | TurboLocalStateRule | TryCatchRule )

SeqRule ::= "seq" Rule Rule ( Rule )* "endseq"

IterateRule ::= "iterate" Rule "enditerate"

TurboCallRule ::= <ID_RULE> "(" ( Term ( "," Term )* )? ")"

TurboReturnRule ::= ( LocationTerm | VariableTerm ) "<-"
TurboCallRule

TurboLocalStateRule ::= DynamicFunction "[" Rule "]" (
DynamicFunction "[" Rule "]" )* Rule

TryCatchRule ::= "try" Rule "catch" ( LocationTerm | VariableTerm
) ( "," ( LocationTerm | VariableTerm ) )* Rule

DerivedRule ::= ( BasicDerivedRule | TurboDerivedRule )

BasicDerivedRule ::= CaseRule

CaseRule ::= "switch" Term ( "case" Term ":" Rule )+ ( "otherwise"
Rule )? "endswitch"

TurboDerivedRule ::= ( RecursiveWhileRule | IterativeWhileRule )

RecursiveWhileRule ::= "whilerec" Term "do" Rule

IterativeWhileRule ::= "while" Term "do" Rule

Final terminals

ID_AGENT ::= <ID_FUNCTION>

<ID_VARIABLE> ::= "$" <LETTER> ( <LETTER> | <DIGIT> )*

<ID_ENUM> ::= [ "A" - "Z" ] [ "A" - "Z" ] ( "_" | [ "A" - "Z" ] |
<DIGIT> )*>



<ID_DOMAIN> ::= [ "A" - "Z" ] ( "_" | [ "a" - "z" ] | <DIGIT>)* >

<ID_RULE> ::= "r_" ( <LETTER> | <DIGIT> )+

<ID_FUNCTION> ::= [ "a" - "z" ] ( <LETTER> | <DIGIT> )*

<ID> ::= <LETTER> ( <LETTER> | <DIGIT> )*

<MOD_ID> ::= ( <LETTER> | "." | "\" | "//" ) ( <LETTER> | <DIGIT>
| "." | "\" | "//" | ":" )*

<NUMBER> ::= ( <DIGIT> )+

<REAL_NUMBER> ::= ( <DIGIT> )+ "." ( <DIGIT> )+

<COMPLEX_NUMBER> ::= ( ( ["+", "-"] )? ( <DIGIT> )+ ( "." (
<DIGIT> )+ )? )? ( ["+", "-"] )? "i" ( ( <DIGIT> )+ ( "." (
<DIGIT> )+ )? )?

<CHAR_LITERAL> ::= " ’ " ( ( ~[ " ’ ", "\\", "\n", "\r" ] ) | (
"\\" ( [ "n", "t", "b", "r", "f", "\\", " ’ ", "\"" ]

| [ "0"-"7" ] ( [ "0"-"7" ] )?
| [ "0"-"3" ] [ "0"-"7" ] [ "0"-"7" ] ) ) )* " ’ "

<STRING_LITERAL> ::= " \" " ( ( ~[ " \" ", "\\", "\n", "\r" ] )
| ( "\\" ( [ "n", "t", "b", "r", "f", "\\", " ’ ", "\"" ] |
["0"-"7" ] ( [ "0"-"7" ] )? | [ "0"-"3" ] [ "0"-"7" ] [ "0"-"7"]
) ) )* " \" "

<LETTER> ::= [ "_" , "a" - "z" , "A" - "Z" ]

<DIGIT> ::= [ "0" - "9" ]


